# IMPLEMENTATION OF A FLUID-STRUCTURE INTERACTION FORMULATION USING MSC/NASTRAN 

S. S. Lee, M. C. Kim, and D. R. Williamson ${ }^{1}$

The Aerospace Corporation
El Segundo, California 90245


#### Abstract

A fluid-structure interaction formulation has been developed previously for incompressible fluids with a free surface. The formulation involves a series of transformations for the coupled fluid-structure equation, which is originally nonsymmetric. The singularity of the fluid inertance matrix is removed by eliminating the rigid body slosh mode in the transformations, and the combined fluid-structure equation is made symmetric. In this paper, a DMAP procedure which implements the formulation is developed using MSC/NASTRAN.


[^0]
## INTRODUCTION

Since the early days of its development, the finite element method has been applied to the fluid-structure interaction problems [1][2][3]. Various NASTRAN programs have provided some form of fluid-structure interaction analysis capability [4][5][6]. These allowed an analyst to study fluid-structure interaction problems without a great effort.

In early 1970s, MSC/NASTRAN offered two approaches for fluid-structure interaction problems. The first is called the virtual mass method, which can be applied to incompressible fluids, with the fluid being internal or external to a structure. However, the gravity effect could not be applied. The second method is applied to axisymmetric problems [2], and can accommodate the gravity effect. For a while, general three-dimensional fluid elements were not available. Special programs or user-developed elements had to be used [7]. As a way to get around this problem, an approach to use the analogy with heat conduction was proposed for general three-dimensional problems [3]. When the general three dimensional fluid elements became finally available in MSC/NASTRAN, it became widely accepted as a way to analyze fluid-structure interaction problems. Also a specialized program has been developed that facilitates a general three dimensional fluid-structure interaction analysis in MSC/NASTRAN [8].

For most engineering applications, the usual assumption of inviscid irrotational incompressible fluid is adequate in characterizing fluid influence on the structure. Therefore, we adopt the same assumptions in this paper. As for the typical fluid-structure interaction formulation, the coupled fluid-structure equation is not symmetric in its original form. Previously, a new formulation has been derived, in which a series of transformations turns the nonsymmetric coupled fluid-structure equation into a symmetric form [9]. The key step in the transformations is the use of slosh modes. Due to the assumption of fluid incompressibility, the fluid inertance matrix is singular, and yields one rigid body slosh mode. To remove the singularity, a constraint equation has been usually imposed in other formulations $[3][7][10]$. In the new formulation, the singularity is removed by eliminating the rigid body slosh mode. In the process, additional mass and stiffness terms that augment the original structure matrices are produced. These explicit terms, which are hidden in the original nonsymmetric equation, allow assessment of the relative importance of the slosh modes. It should be noted that in the past no particular attention has been paid to the rigid body slosh mode. Most of the references did not include the rigid body slosh mode when presenting slosh mode analysis results, let alone discussing its importance.

In this paper, a Direct Matrix Abstraction Program (DMAP) procedure was developed to implement the new formulation using MSC/NASTRAN. Various fluid-structure interaction formulations are briefly mentioned only for comparison purpose. Comparison with existing fluid-structure interaction formulations shows that the implementation is efficient and concise.

## A FLUID-STRUCTURE INTERACTION FORMULATION

Governing equations for incompressible, inviscid, irrotational fluids are discretized into a matrix equation using the finite element method, in which fluid pressures are the dependent variables [11] [12]. Also developed is a matrix equation for the structure in a usual fashion using the finite element method. The small motion assumptions for both the fluid and the structure have been used in the process, and the matching mesh compatibility is assumed at the wetted interface between the fluid and the structure. In the following development, a free surface is assumed to exist, and the surface tension is neglected. The gravity constant is assumed to be nonzero.

## Fluid Model

Under the assumptions discussed above, the matrix equation for an incompressible fluid [3] is given by

$$
\left[\begin{array}{ccc}
\bar{L}_{f f} & \bar{L}_{f w} & \bar{L}_{f i}  \tag{1}\\
\bar{L}_{w f} & \bar{L}_{w w} & \bar{L}_{w i} \\
\bar{L}_{i f} & \bar{L}_{i w} & \bar{L}_{i i}
\end{array}\right]\left\{\begin{array}{c}
p_{f} \\
p_{w} \\
p_{i}
\end{array}\right\}=\left\{\begin{array}{c}
P_{f} \\
P_{w} \\
0
\end{array}\right\}
$$

where $\bar{L}$ is the fluid inertance matrix, $p$ the pressure, and $P$ the generalized force imparted to the fluid. The pressure partitions $p_{f}, p_{w}$, and $p_{i}$ correspond to the free surface, the (wetted) structural interface, and the internal fluid pressure sets, respectively. Due to the uniform gravity assumption, the generalized force on the fluid interior is zero.

In the fluid formulation, pressure variables are defined at undisturbed positions in space. When a free surface is displaced, the free surface pressure $p_{f}$ is given in terms of the outward normal displacement of the fluid free surface $u_{f}$

$$
\begin{equation*}
p_{f}=\rho g u_{f} \tag{2}
\end{equation*}
$$

where $\rho$ is the density of the fluid, and $g$ the nonzero gravity constant. By multiplying the above equation with the free surface area matrix $A_{f f}$, which is a symmetric positive definite matrix, the equation becomes

$$
\begin{equation*}
K_{f f} u_{f}-A_{f f} p_{f}=0 \tag{3}
\end{equation*}
$$

where the free surface stiffness matrix $K_{f f}$ is defined by

$$
\begin{equation*}
K_{f f} \equiv \rho g A_{f f} \tag{4}
\end{equation*}
$$

The free surface stiffness matrix $K_{f f}$ is symmetric positive definite, and therefore, its inverse exists. Multiplying equation (3) by $A_{f f}^{T} K_{f f}^{-1}$ and twice differentiating with respect to time yields

$$
\begin{equation*}
A_{f f}^{T} \ddot{u}_{f}-M_{f f} \ddot{p}_{f}=0 \tag{5}
\end{equation*}
$$

where the free surface "mass" matrix $M_{f f}$ is defined to be

$$
\begin{equation*}
M_{f f} \equiv A_{f f}^{T} K_{f f}^{-1} A_{f f} \tag{6}
\end{equation*}
$$

which is also symmetric positive definite. The double dot denotes differentiation twice with respect to time, and $T$ denotes the transpose of a matrix. The fluid free surface mass matrix is used in calculating the slosh modes of the fluid in a rigid structure.

The generalized force in the fluid at the free surface is related to the outward normal displacement through the free surface area matrix and the flow relationship

$$
\begin{equation*}
P_{f}=-A_{f f}^{T} \ddot{u}_{f} \tag{7}
\end{equation*}
$$

Static condensation of the interior pressure $p_{i}$ from equation (1) as well as incorporating equations (5) and (7) yields the final fluid equation

$$
\left[\begin{array}{cc}
M_{f f} & 0  \tag{8}\\
0 & 0
\end{array}\right]\left\{\begin{array}{c}
\ddot{p}_{f} \\
\ddot{p}_{w}
\end{array}\right\}+\left[\begin{array}{cc}
L_{f f} & L_{f w} \\
L_{w f} & L_{w w}
\end{array}\right]\left\{\begin{array}{c}
p_{f} \\
p_{w}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
P_{w}
\end{array}\right\}
$$

## Structure Model

The structural model is also partitioned;

$$
\left[\begin{array}{ll}
M_{b b} & M_{b c}  \tag{9}\\
M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{c}
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{ll}
K_{b b} & K_{b c} \\
K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{l}
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{l}
F_{b} \\
F_{c}
\end{array}\right\}
$$

where $M$ and $K$ are the mass and stiffness matrices, $F$ the applied forces, and $u_{b}$ and $u_{c}$ the structural displacement sets corresponding to the fluid interface displacements and the rest of the displacements, respectively. The damping matrix is not considered in this formulation.

## Coupled Fluid and Structure Model

The generalized force in the fluid at the structure interface is related to the structure displacement through the wetted surface area matrix and the flow relationship [4]

$$
\begin{equation*}
P_{w}=-A_{b w}^{T} \ddot{u}_{b} \tag{10}
\end{equation*}
$$

where $A_{b w}$ is the wetted surface area matrix. In turn, the fluid applies forces over the structure surface area

$$
\begin{equation*}
F_{b}=A_{b w} p_{w}-K_{g} u_{b} \tag{11}
\end{equation*}
$$

where $K_{g}$ is the gravity stiffness matrix [7]. For most engineering applications, $K_{g}$ is much smaller than $K_{b b}$, and is neglected in the following development. Combining the fluid and structure equations (8) and (9) into a matrix equation incorporating equations (10) and (11) yields

$$
\left[\begin{array}{cccc}
M_{f f} & 0 & 0 & 0  \tag{12}\\
0 & 0 & A_{b w}^{T} & 0 \\
0 & 0 & M_{b b} & M_{b c} \\
0 & 0 & M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{c}
\ddot{p}_{f} \\
\ddot{p}_{w} \\
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{cccc}
L_{f f} & L_{f w} & 0 & 0 \\
L_{w f} & L_{w w} & 0 & 0 \\
0 & -A_{b w} & K_{b b} & K_{b c} \\
0 & 0 & K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{c}
p_{f} \\
p_{w} \\
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
0 \\
0 \\
F_{c}
\end{array}\right\}
$$

It is obvious that the coupled fluid-structure equation is nonsymmetric. Solving the second row in equation (12) for $p_{w}$, and substituting back into the equation yields

$$
\left[\begin{array}{ccc}
M_{f f} & -A_{b f}^{T} & 0  \tag{13}\\
0 & M_{b b}+m_{b b} & M_{b c} \\
0 & M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{c}
\ddot{p}_{f} \\
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{ccc}
\tilde{L}_{f f} & 0 & 0 \\
A_{b f} & K_{b b} & K_{b c} \\
0 & K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{c}
p_{f} \\
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
0 \\
F_{c}
\end{array}\right\}
$$

where

$$
\begin{gather*}
m_{b b} \equiv A_{b w} L_{w w}^{-1} A_{b w}^{T}  \tag{14}\\
\tilde{L}_{f f} \equiv L_{f f}-L_{f w} L_{w w}^{-1} L_{w f}  \tag{15}\\
A_{b f} \equiv A_{b w} L_{w w}^{-1} L_{w f} \tag{16}
\end{gather*}
$$

The matrix $m_{b b}$ is usually called the added mass of the fluid to the structure. Equation (13) is still nonsymmetric, but can be transformed into a symmetric form through a series of transformations. The key step in the transformations is the use of the slosh modes. Solving the eigenvalue problem for the first row in equation (13) after putting $\left\{\ddot{u}_{b}\right\}=0$ (rigid container), i.e.

$$
\begin{equation*}
M_{f f} \ddot{p}_{f}+\tilde{L}_{f f} p_{f}=0 \tag{17}
\end{equation*}
$$

slosh modes $\phi_{f f}$ are obtained, where

$$
\begin{gather*}
p_{f} \equiv \phi_{f f} q_{f}  \tag{18}\\
I_{f f} \equiv \phi_{f f}^{T} M_{f f} \phi_{f f}  \tag{19}\\
W_{f f} \equiv \phi_{f f}^{T} \tilde{L}_{f f} \phi_{f f} \tag{20}
\end{gather*}
$$

where $q_{f}$ is a vector of generalized coordinates, $I_{f f}$ is an identity matrix, and $W_{f f}$ is a diagonal matrix with the circular frequency squared on the diagonal. All slosh modes are retained; i.e., no modal truncation is used in the eigensolution. Using the slosh modes, equation (13) can be converted into the following form

$$
\left[\begin{array}{ccc}
I_{f f} & -\phi_{f f}^{T} A_{b f}^{T} & 0  \tag{21}\\
0 & M_{b b}+m_{b b} & M_{b c} \\
0 & M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{c}
\ddot{q}_{f} \\
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{ccc}
W_{f f} & 0 & 0 \\
A_{b f} \phi_{f f} & K_{b b} & K_{b c} \\
0 & K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{c}
q_{f} \\
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
0 \\
F_{c}
\end{array}\right\}
$$

The matrix $\tilde{L}_{f f}$ has a rank deficiency of one, due to the fact that the incompressible fluid volume does not change under the static uniform pressure; i.e. from equation (17), we have $\tilde{L}_{f f}\{1\}=0$, where $\{1\}$ is a vector of 1 's. A well-known theorem in linear algebra states that for a symmetric matrix $A$, the equation $A x=0$ has a nontrivial solution $x \neq 0$ if and only if det $A=0$ [13]. Hence, $\tilde{L}_{f f}$ is singular, and there exists a rigid body (zero frequency) slosh mode. Partitioning the generalized coordinates $q_{f}$ into $q_{r}$ (corresponding to the zero frequency slosh mode) and $q_{n}$ (corresponding to the rest of slosh modes) we obtain

$$
W_{f f}=\left[\begin{array}{cc}
W_{r r} & 0  \tag{22}\\
0 & W_{n n}
\end{array}\right]
$$

With $W_{r r}=0$, equation (21) becomes

$$
\left[\begin{array}{cccc}
I_{r r} & 0 & -A_{b r}^{T} & 0  \tag{23}\\
0 & I_{n n} & -A_{b n}^{T} & 0 \\
0 & 0 & M_{b b}+m_{b b} & M_{b c} \\
0 & 0 & M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{c}
\ddot{q}_{r} \\
\ddot{q}_{n} \\
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{cccc}
0 & 0 & 0 & 0 \\
0 & W_{n n} & 0 & 0 \\
A_{b r} & A_{b n} & K_{b b} & K_{b c} \\
0 & 0 & K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{c}
q_{r} \\
q_{n} \\
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
0 \\
0 \\
F_{c}
\end{array}\right\}
$$

where $A_{b f} \phi_{f f} \equiv\left[A_{b r} A_{b n}\right]$ and in fact $I_{r r}=1$. The term $A_{b r}$ represents the shape that the wetted part of the container takes in order to yield the rigid body slosh mode. In other words, the free surface of the fluid rises uniformly when the container is statically squeezed into this shape. From the first row of equation (23), the generalized coordinate corresponding to the rigid body slosh mode is obtained in terms of the wetted structure displacement

$$
\begin{equation*}
q_{r}=I_{r r}^{-1} A_{b r}^{T} u_{b} \tag{24}
\end{equation*}
$$

Using equation (24), equation (23) reduces to the form

$$
\left[\begin{array}{ccc}
I_{n n} & -A_{b n}^{T} & 0  \tag{25}\\
0 & M_{b b}+m_{b b} & M_{b c} \\
0 & M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{l}
\ddot{q}_{n} \\
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{ccc}
W_{n n} & 0 & 0 \\
A_{b n} & K_{b b}+k_{b b} & K_{b c} \\
0 & K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{l}
q_{n} \\
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
0 \\
F_{c}
\end{array}\right\}
$$

where

$$
\begin{equation*}
k_{b b} \equiv A_{b r} I_{r r}^{-1} A_{b r}^{T} \tag{26}
\end{equation*}
$$

The matrix $W_{n n}$ is a symmetric positive definite matrix, and hence its inverse exists. Therefore, the final form of the coupled fluid-structure equation is obtained by multiplying the first row of equation (25) by $-A_{b n} W_{n n}^{-1}$, and adding the resulting equation to the second row, and multiplying the first row by $W_{n n}^{-1}$,

$$
\left[\begin{array}{ccc}
W_{n n}^{-1} & -W_{n n}^{-1} A_{b n}^{T} & 0  \tag{27}\\
-A_{b n} W_{n n}^{-1} & M_{b b}+m_{b b}+\mu_{b b} & M_{b c} \\
0 & M_{c b} & M_{c c}
\end{array}\right]\left\{\begin{array}{l}
\ddot{q}_{n} \\
\ddot{u}_{b} \\
\ddot{u}_{c}
\end{array}\right\}+\left[\begin{array}{ccc}
I_{n n} & 0 & 0 \\
0 & K_{b b}+k_{b b} & K_{b c} \\
0 & K_{c b} & K_{c c}
\end{array}\right]\left\{\begin{array}{l}
q_{n} \\
u_{b} \\
u_{c}
\end{array}\right\}=\left\{\begin{array}{c}
0 \\
0 \\
F_{c}
\end{array}\right\}
$$

where

$$
\begin{equation*}
\mu_{b b} \equiv A_{b n} W_{n n}^{-1} A_{b n}^{T} \tag{28}
\end{equation*}
$$

Equation (27) is now symmetric, and no approximation was used in the series of transformations leading to it from the original nonsymmetric equation (12). The symmetric form allows it to be readily converted into a Craig-Bampton component dynamic model [14].

## IMPLEMENTATION IN MSC/NASTRAN

The procedure for the fluid-structure interaction formulation is implemented using MSC/NASTRAN DMAP (V68.2), whose listing is included in Appendix. The DMAP program consists of alters to the normal modes calculation in subDMAP SEMODES (SOL 103) and a new subDMAP called MODEFSR. The fluid-structure model must be defined as the residual structure. Since the fluid portion of the model has a different frequency range than the structural portion, a different eigenvalue extraction method should be specified for each portion. This is achieved by using the METHOD(FLUID) command for the fluid portion and METHOD(STRUCTURE) for the structural portion in the Case Control Section.

The fluid model can be generated by using existing MSC/NASTRAN three-dimensional elements: CHEXA, CPENTA, and CTETRA. Fluid grid points belonging to the free and wetted surfaces are defined by MSC/NASTRAN's c-set and b-set, respectively. The interior fluid grid points belong to the o-set, and they are eliminated by the static condensation process, resulting in the fluid inertance matrix in equation (8).

The interface between the fluid and the structure is modeled so that the grid points of the fluid are coincident with those of the structure (matching mesh). Hence, the structure model includes the container structure and a set of artificial grids to represent the free surface. The artificial grids are needed to generate the free surface area matrix, and only the outward normal degrees of freedom (DOF) are assigned to the q-set. Also, with the use of PLOTEL cards, the slosh modes can be plotted. The wetted and non-wetted structure DOF are assigned to the b-set and c-set, respectively.

Definition of fluid-structure interface is determined automatically for the wetted and free surfaces by entering IDENT in the INTER field of the ACMODL Bulk Data entry and by defining SET1 entry for the fluid and structural grid points. The total free surface area in the basic coordinate system is printed out under a DMAP INFORMATION MESSAGE 9055. Once the free surface area matrix is obtained, $M_{f f}$ is generated by scaling the area matrix. Then $\tilde{L}_{f f}$ is calculated from equation (15), and the slosh modes are calculated from the eigenvalue problem by calling subDMAP MODERS.

The rest of the procedure follows the formulation in a straightforward manner. The DMAP listing includes corresponding equation numbers in the previous section. In the end, all components in equation (27) are assembled, and the eigenproblem for the coupled fluid-structure is solved by a final call to subDMAP MODERS. An example of a hydroelastic analysis using the attached DMAP program is provided [15].

## COMPARISON WITH OTHER FORMULATIONS

## Nonsymmetric Formulation

The original coupled fluid-structure equation is nonsymmetric. One finite element program uses a special eigenvalue solver in its direct approach to solve the nonsymmetric equation [4]. This brute force approach gives correct results, but does not provide insights into the fluid-structure interaction problems. By deriving explicit components of interaction, the approach presented here gives better understanding on how the added mass as well as the slosh modes affect the structure response.

A more important point of using the transformations in the approach is that the final symmetric equation tells us much more. For example, the slosh mode frequencies of the coupled fluid-structure equations are less than or equal to the corresponding frequencies for the rigid tank, from the inclusion theorem in algebraic eigenvalue theory [16]. In addition, the slosh modes further lower the structure mode frequencies when compared with the corresponding frequencies with only added mass and stiffness components in the final symmetric fluid-structure equation.

## Constraint Equation

When the pressure formulation for the fluid is converted into a displacement formulation, the constraint of volume constancy has to be applied. This is referred to as the multipoint constraint equation. It relates the free surface and wetted surface displacements $[3][7][10]$. The multipoint constraint equation eliminates the rigid body slosh mode, and in the process distributes the mass and stiffness associated with a reference displacement (one of the free surface displacements is usually selected) onto the rest of the displacements. These are related to the terms given in equations (26) and (28).

The multipoint constraint equation is cumbersome to generate. The transformation procedure described in the preceding section accomplishes the same task in a more concise manner than the multipoint constraint equation. In some finite element programs, the single point constraint equation is recommended to avoid this cumbersomeness. The single point constraint also eliminates the rigid body slosh mode, but neglects the terms in equations (26) and (28).

## Area Matrix Calculation

The area matrix calculation involves applying a pressure profile on the surface elements and calculating the resulting loads [10]. For a large fluid-structure interaction problem, this means multiple static solution runs. Also, scores of PLOAD4 cards have to be generated. In this implementation, MSC/NASTRAN DMAP module ACMG is used to compute the area matrix when the fluid and structure have identical
meshes at the interface. Using the ACMG module eliminates the need for PLOAD4 cards and multiple static solution runs.

## Slosh Modes Calculation

In MSC's fluid-structure interaction formulation [17][18], artificial springs are used in defining the free surface potential. In this formulation, slosh modes are calculated using the free surface area matrix. The formulation yields more accurate slosh mode frequencies than the formulation using the artificial springs. It also eliminates the need to generate the CELAS entries.

## CONCLUDING REMARKS

The implementation of a fluid-structure interaction formulation using MSC/NASTRAN is presented in this paper. The formulation is for incompressible fluids with a free surface. The formulation involves a series of transformations for the nonsymmetric coupled fluid-structure equation which is converted into a symmetric form with no approximations. The key step in the transformations is the use of the rigid body slosh mode, and in the process additional stiffness and mass terms are produced that augment the original structure matrices. These explicit terms, which are hidden in the original unsymmetric coupled fluid-structure equations, allow assessment of the relative importance of the slosh modes. The procedure offered another way of removing the singularity of the fluid inertance matrix, which was usually handled by multipoint or single point constraint equations in finite element programs. It is hoped that analysts can use the attached DMAP listing in their effort to learn hydroelastic analysis approaches.

## REFERENCES

[1]. Belytschko, T., "Methods and Programs for Analysis of Fluid-Structure Systems," Nuclear Engineering and Design, 42, 41-52, 1977.
[2]. MacNeal, R. H., Editor, "The NASTRAN Theoretical Manual," NASA SP-221(01), 1972.
[3]. Coppolino, R. N., "A Numerically Efficient Finite Element Hydroelastic Analysis, Volume I: Theory and Results," NASA CR-2662, 1976.
[4]. Univeral Analytics, Inc., "UAI/NASTRAN User's Guide," 1995.
[5]. Computerized Structural Analysis and Research Corporation, "CSA/NASTRAN User's Manual," 1994.
[6]. The MacNeal-Schwendler Corporation, " MSC/NASTRAN Reference Manual V68," in MSC/ NASTRAN Encyclopedia, Second Edition, 1995.
[7]. Universal Analytics, Inc., "Final Report NASTRAN Hydroelastic Modal Studies, Volume I Introduction, Theory, and Results," NASA CR-150393, 1977.
[8]. Structural Dynamics Research Corporation, "Hydro User's Guide, Version 1," 1992.
[9]. M. C. Kim and D. R. Williamson, "A Fluid-Structure Interaction Formulation for Fluids With Free Surface," Proceedings of the 6th AIAA/NASA/ISSMO Symposium on Multidisciplinary Analysis and Optimization, 1763-1772, 1996.
[10]. Flanigan, C. C., "Hydroelastic Analysis of Incompressible Fluids Using SDRC I-DEAS and MSC/NASTRAN," presented at 1992 MSA/NASTRAN World Users Conference, Dearborn, Michigan, 18-22 May, 1992.
[11]. Chargin, M., and Gartmeier, O., "A Finite Element Procedure for Calculating Fluid-Structure Interaction Using MSC/ NASTRAN," NASA TM-102857, 1990.
[12]. Zienkiewicz, O. C., The Finite ELement Method in Engineering Science, McGraw-Hill, London, 1971.
[13]. Shilov, G. E., Linear Algebra, Dover Publications, Inc., New York, 1977.
[14]. Craig, R. R., Jr., and Bampton, M. C. C., "Coupling of Substructures for Dynamic Analysis," AIAA Journal, 6(7), 1313-1319, 1968.
[15]. Kim, M. C. and Lee, S. S., "Hydroelastic Analysis of a Rectangular Tank," 1997 MSC Aerospace Users' Conference, Newport Beach, California, November 17-20, 1997.
[16]. Wilkinson, J. H., The Algebraic Eigenvalue Problem, Clarendon Press, Oxford, 1965.
[17]. Beall, G., "Acoustic Analysis Application note," The First India MSC Users' Conference Proceedings, December 1992.
[18]. Beall, G., "Acoustic Analysis Makes Sloshing No Problem," MSC/WORLD Volume III, No. 1, January 1993.

## APPENDIX. DMAP LISTING (MSC/NASTRAN V68.2)

```
$
$ 'hydro_msc.dmap'
$
COMPILE SEMODES SOUIN=MSCSOU
$
ALTER 86,86 $ Remove CALL MODERS
$
ALTER 96 $ Insert after NOSASET=NOASET-NOFASET
$
CALL MODEFSR MR,USET,DM,CASES,DYNAMICS,MMAA,MKAA,GPLS,SILS,EED,
    EQEXINS,VAFS,AGG/
    PHA,LAMA/
    NORSET/NOLSET/ASING/NOASET/
    NOFASET/NOSASET/METH/TRUE/NOQSET $
$
ENDALTER $
$
COMPILE MODEFSR
$
SUBDMAP MODEFSR MR,USET,DM,CASES,DYNAMICS,MMAA,MKAA,GPLS,SILS,EED,
            EQEXINS,VAFS,AGG/
                                PHA,LAMA/
                                NORSET/NOLSET/ASING/NOASET/
                                NOFASET/NOSASET/METH/NASOUT/NOQSET $
$
TYPE DB KFAA,KSAA,MFAA,MSAA $ SCRATCH
TYPE PARM,,LOGICAL,N,FS,NASOUT $
TYPE PARM,,I,N,NORSET,NOLSET,NOA,ASING,NOFASET,NOSASET,NOASET,
                                    NSEIG=-1,NFEIG=-1,NFEIG1,NFSEIG , METH , METHF , NOQSET ,
                                    NOFRSET=-1,NOSRSET=-1,
                                    NOFCSET=-1, NOFBSET=-1,
                            NOSCSET=-1,NOSBSET=-1,NOSQSET=-1 $
TYPE PARM,,RS,Y,GRAVITY=1.0,DENSITY=1.0 $
TYPE PARM,,CS,N,RHOG1 $
$
PARAML CASES//'DTI'/1/226//S,N,METHF $
$
RHOG1=CMPLX(1.0/(DENSITY*GRAVITY))
$
FS=(NOSASET>0 AND NOFASET>0) $ BOTH FLUID AND STRUCTURE EXIST
$
IF ( METH=0 AND METHF=0 ) THEN $
    MESSAGE //
    , USER FATAL MESSAGE **** (MODEFSR) - NO METHOD COMMAND WAS '/
    'SPECIFIED IN THE CASE CONTROL SECTION.' $
    CALL PRTSUM /RSTJUNK $
    EXIT $
ENDIF $
$
IF ( NORSET > -1 ) THEN $
    NOFRSET=0 $
```

```
    IF ( FS ) THEN $
    UPARTN USET,VAFS/VRFS,,,/'A'/'R'/'L'/1 $
    PARAML VRFS//'TRAILER'/5/S,N,NOFRSET $
    PARAML VRFS//'TRAILER'/4/S,N,PREC $
    NOFRSET=NOFRSET/PREC $ NO. OF FLUID DOFS IN R-SET
    ENDIF $
    NOSRSET=NORSET-NOFRSET $ NO. OF STRUCTURE DOFS IN R-SET
    NOFRSET=-1 $ R-set not allowed for fluid, ignore it
    IF ( NOSRSET<=O ) NOSRSET=-1 $
ENDIF $
$
IF ( FS ) THEN $
    UPARTN USET,VAFS/VCFS,,,/'A'/'C'/'L'/1 $
    PARAML VCFS//'TRAILER'/5/S,N,NOFCSET $
    PARAML VCFS//'TRAILER'/4/S,N,PREC $
    NOFCSET=NOFCSET/PREC $ NO. OF FLUID DOFS IN C-SET (Free Surface)
    NOFBSET=NOFASET-NOFCSET $ NO. OF FLUID DOFS IN B-SET (Wetted Surface)
    NOSBSET=NOFBSET $ NO. OF STRUCTURE DOFS IN B-SET (Wetted Surface)
    NOSQSET=NOQSET $ NO. OF STRUCTURE DOFS IN Q-SET (Slosh modes)
    NOSCSET=NOSASET-NOSBSET-NOSQSET $ NO. OF STRUCTURE DOFS IN C-SET (Free Surface)
    IF ( NOSCSET<=O ) NOSCSET=-1 $
    PRTPARM //O $ DISPLAY SIZES
ENDIF $
$
VEC USET/VACMPB/'A'/'COMP'/'B' $
VEC USET/VACMPQ/'A'/'COMP'/'Q' $
IF ( FS ) THEN $
    PARTN VACMPB,,VAFS/VACMPBS,VACMPBF,,/1 $
    PARTN VACMPQ,,VAFS/VACMPQS,,,/1 $
ELSE $
    EQUIVX VACMPB/VACMPBS/-1 $
    EQUIVX VACMPQ/VACMPQS/-1 $
ENDIF $
$
$ PARTITION THE K AND M MATRICES AND INTO FLUID AND STRUCTURE COMPONENTS
$
IF ( NOFASET=0 ) THEN $
    EQUIVX MKAA/KSAA/ALWAYS $
    EQUIVX MMAA/MSAA/ALWAYS $
ELSE IF ( NOSASET=O ) THEN $
    EQUIVX MKAA/KFAA/ALWAYS $
    EQUIVX MMAA/MFAA/ALWAYS $
ELSE $
    PARTN MKAA,VAFS,/KSAA,,,KFAA $
    PARTN MMAA,VAFS,/MSAA,,,MFAA $
ENDIF $
$
IF ( NOSQSET > -1 ) THEN $
    PARTN MSAA,VACMPQS,/MSTT,,,MSQQ/-1 $
    PARTN KSAA,VACMPQS,/KSTT,,,KSQQ/-1 $
ELSE $
    MESSAGE //
    ' USER FATAL MESSAGE **** (MODEFSR) - NO STRUCTURE Q-SET '/
    'SPECIFIED FOR FLUID SLOSH MODES.' $
```

CALL PRTSUM /RSTJUNK \$
EXIT \$

## ENDIF \$

\$
IF ( NOSCSET > -1 ) THEN \$
PARTN VACMPBS, ,VACMPQS/VTCMPBS,,, /1 \$
PARTN MSTT,VTCMPBS,/MSBB,MSCB,MSBC,MSCC/-1 \$
PARTN KSTT, VTCMPBS,/KSBB,KSCB ,KSBC, KSCC/-1 \$
ELSE \$
EQUIVX MSTT/MSBB/ALWAYS \$
EQUIVX KSTT/KSBB/ALWAYS \$
ENDIF \$
\$
IF (METHF>0 AND NOFASET>0) THEN \$
\$
UPARTN USET,AGG/AAA, , /'G'/'A'/'O' \$
PARTN AAA,VAFS,/,,ASAFA,/ \$
PARTN ASAFA,VACMPBF,VACMPQS/,AFF1,ASTFB,/1 \$
IF ( NOSCSET > -1 ) THEN \$
PARTN ASTFB, ,VTCMPBS/,ABW1,,/1 \$
ELSE \$
EQUIVX ASTFB/ABW1/ALWAYS \$
ENDIF \$
ADD ABW1,/ABW/(-1.0,0.0) \$
ADD AFF1,/AFF/(-1.0,0.0) \$
\$
ADD AFF,/CFF/RHOG1 \$
MODTRL CFF////6 \$
\$
PARTN KFAA,VACMPBF,/LFF,LWF,LFW,LWW/-1 \$
MODTRL LWW////6 \$
\$
DECOMP LWW/LLWW,/1 \$
FBS LLWW, ,LWF/LWF1/1 \$
MPYAD LWF,LWF1,LFF/LFFBAR/1/-1///6 \$
\$
MPYAD ABW,LWF1,/ABF \$
\$
TRNSP ABW/ABWT \$
FBS LLWW, ,ABWT/ABWT1/1 \$
MPYAD ABW,ABWT1,/MBB3 \$
\$
\$
CALL MODERS MR,USET,DM, CASES, DYNAMICS, CFF, LFFBAR, GPLS, SILS, EED, EQEXINS, ,/
PHFA, LAMAF/
NOFRSET/NOLSET/'MODES'/ASING/FALSE/ TRUE/TRUE/NOQSET/FALSE \$
PARAML PHFA//'TRAILER'/1/S,N,NFEIG \$
\$
IF ( NOT (NFEIG = NOFCSET) ) THEN \$ MESSAGE //
, USER FATAL MESSAGE **** (MODEFSR) -'/
, NUMBER OF SLOSH MODES '/NFEIG/

```
        ' MUST EQUAL TO THE'/
        , NUMBER OF FLUID FREE SURFACE DOFS '/NOFCSET/ $
        CALL PRTSUM /RSTJUNK $
        EXIT $
    ENDIF $
$
    SMPYAD PHFA,LFFBAR,PHFA,,,/WFF1/3////1 $
    SMPYAD PHFA,CFF,PHFA,,,/IFF1/3////1 $
    MATMOD WFF1,,,,,/WFF,/2////0.000000001 $
    MATMOD IFF1,,,,,/IFF,/2////0.000000001 $
$
    NFEIG1 = NFEIG - 1
    MATGEN, /VRN/6/NFEIG/1/NFEIG1 $ To remove first (0 frequency) mode
$
    PARTN WFF,VRN,/,,,WNN/-1 $
    PARTN IFF,VRN,/IRR,,,INN/-1 $
$
    MPYAD ABF,PHFA,/ABF1 $
    PARTN ABF1,VRN,/ABR,,ABN,/1 $
$
ELSE $
    PURGEX /PHFA,LAMAF,,,/ALWAYS $ EMPTY
ENDIF $ METHF>0 AND NOFASET>0
$
$
IF (METH>0 AND NOSASET>0) THEN $
$
    TRNSP ABR/ABRT $
    SOLVE IRR,/IRRI/3 $
    SMPYAD ABR,IRRI,ABRT,,,/KBB3/3 $
$
    TRNSP ABN/ABNT $
    DECOMP WNN/LWNN,/1 $
    FBS LWNN,,ABNT/ABNT1/1/-1 $
    MPYAD ABN,ABNT1,/MBB5//-1 $
$
    FBS LWNN,,INN/WNNI/1 $
$
    ADD5 KSBB,KBB3,,,/KSBB1/ $
    ADD5 MSBB,MBB3,MBB5,,/MSBB1/ $
$
    MERGE, ,,,INN,VRN,/KSQQ1/-1 $
    MERGE, ,,,WNNI,VRN,/MSQQ1/-1 $
    MERGE, ,ABNT1,,,,VRN/MSBQT1/1 $
    TRNSP MSBQT1/MSBQ1 $
$
    IF ( NOSCSET > -1 ) THEN $
        MERGE KSBB1,KSCB,KSBC,KSCC,VTCMPBS,/KSTT1/-1 $
        MERGE MSBB1,MSCB,MSBC,MSCC,VTCMPBS,/MSTT1/-1 $
        MERGE MSBQT1,,,,VTCMPBS,/MSTQT1/1 $
        TRNSP MSTQT1/MSTQ1 $
    ELSE $
        EQUIVX KSBB1/KSTT1/ALWAYS $
        EQUIVX MSBB1/MSTT1/ALWAYS $
```

```
            EQUIVX MSBQT1/MSTQT1/ALWAYS $
            EQUIVX MSBQ1/MSTQ1/ALWAYS $
    ENDIF $
$
    MERGE KSTT1,,,KSQQ1,VACMPQS,/KSAA1/-1 $
    MERGE MSTT1,MSTQT1,MSTQ1,MSQQ1,VACMPQS,/MSAA1/-1 $
$
    CALL MODERS MR,USET,DM,CASES,DYNAMICS,MSAA1,KSAA1,GPLS,SILS,EED,
                EQEXINS, ,VAFS/
                PHSA,LAMAS/
                NOSRSET/NOLSET/'MODES'/ASING/FALSE/
                FALSE/TRUE/NOQSET/FS $
    PARAML PHSA//'TRAILER'/1/S,N,NSEIG $
$
ELSE $
    PURGEX /PHSA,LAMAS,,,/ALWAYS $ EMPTY
ENDIF $ METH>0 AND NOSASET>0
$
IF ( FS ) THEN $
$
    IF (NSEIG>0 AND NFEIG>0) THEN $
$
    NFSEIG=NSEIG+1
    MATGEN ,/CL1/6/NFSEIG/1/NSEIG $
    MATGEN ,/CL2/6/5/2/2/1 $
    MATGEN ,/CL3/6/3/1/1/1 $
$
    PARTN PHSA,,VACMPQS/PHST,PHSQ,,/1 $
    MERGE, ,,PHST,,CL1,/PHST1/1 $
    MATGEN ,/I11/1/1 $
    PARTN PHSQ,,VRN/,PHSN,,/1 $
    MERGE I11,,,PHSN,CL1,VRN/PHSQ3/1 $
$
    MATMOD AFF,,,,,/AFFMAX,/6 $
    MATGEN ,/UNIT/6/NOFCSET/0/NOFCSET $
    ADD UNIT,AFFMAX/AFFMAX1///2 $
    MATMOD AFFMAX1,,,,,/AFFMAX1D,/28 $
    MPYAD AFF,AFFMAX1D,/AFFX1 $ EQ.
    MATMOD AFFX1,,,,,/AFFX,/2////0.9999 $
    SMPYAD AFFX,PHFA,PHSQ3,,,/PHSQ2/3 $
    ADD PHSQ2,/PHSQ1/RHOG1 $
$
    MERGE PHST1,PHSQ1,,,,VACMPQS/PHSA1/1 $
    MERGE PHSA1,,,PHFA,,VAFS/PHA/1 $
$
    LAMX, ,LAMAF/LMATF/-1 $
    LAMX, ,LAMAS/LMATS/-1 $
    PARTN LMATF,,VRN/LMATFR,,,/1 $
    MERGE LMATFR,LMATS,,,,CL1/LMATFS/1 $
    PARTN LMATFS,CL2,/,,LMATFS1,/1 $
    MERGE LMATFS1,,,,CL3,/LMATFS2/1 $
    TRNSP LMATFS2/LMATFS3 $
    LAMX LMATFS3,/LAMA $
    OFP LAMA//$
```

```
    ENDIF $
$
ENDIF $ FS
$
PRTPARM //O $ DISPLAY SIZES
$
RETURN $
END $ MODEFSR
```


[^0]:    ${ }^{1}$ Current address: TRW Space Systems, Redondo Beach, California 90277.

